# Contributor: Ionut Biru <ibiru@archlinux.org>

pkgname=gnome-tweak-tool
pkgver=3.28.0
pkgrel=1
pkgdesc="Customize advanced GNOME 3 options"
url="https://wiki.gnome.org/Apps/GnomeTweakTool"
arch=(any)
license=(GPL)
depends=(gnome-settings-daemon python-gobject nautilus)
makedepends=(git meson)
groups=(gnome-extra)
_commit=9e4143d786c29c88d289b3d5755b51a47bdb650f  # tags/3.28.0^0
source=("git+https://git.gnome.org/browse/gnome-tweak-tool#commit=$_commit")
sha256sums=('SKIP')

pkgver() {
  cd $pkgname
  git describe --tags | sed 's/-/+/g'
}

build() {
  arch-meson $pkgname build
  ninja -C build
}

check() {
  cd build
  meson test
}

package() {
  DESTDIR="$pkgdir" ninja -C build install
}
